

Zaawansowane Systemy Baz Danych – ZSBD

**Bazy danych dokumentów XML
wykład 2 – XQuery**

**Wykład przygotował:
Krzysztof Jankiewicz**

Bazy danych dokumentów XML – wykład 2 – XQuery

Drugi wykład z baz danych dokumentów XML zostanie poświęcony językowi zapytań XQuery. W roku 2005 język XQuery stał się standardem zatwierdzonym przez W3C.



Wymagania W3C dotyczące języka zapytań wymagania ogólne

- Ogólne wymagania W3C dotyczące języka zapytań dotyczą następujących zagadnień:
 - Składnia języka
 - Deklaratywność
 - Niezależność od protokołów
 - Warunki wystąpienia błędów
 - Modyfikacje

Bazy danych dokumentów XML – wykład 2 – XQuery (2)

W dokumencie "XML Query Requirements" organizacja W3C zebrała podstawowe wymagania dotyczące języków zapytań przeznaczonych do przetwarzania dokumentów XML.

Wymagania te dotyczą następujących zagadnień: składnia języka, deklaratywność, niezależność od protokołów, warunki wystąpienia błędów, modyfikacje.

I tak dla przykładu wymagania dotyczące składni języka są następujące:

- język zapytań **może** mieć więcej niż jedną składnię,
- **musi** być zrozumiały w czytaniu i pisaniu przez człowieka,
- co najmniej jedna składnia **musi** być wyrażona w formacie XML.

Ponadto język przeznaczony do przetwarzania dokumentów XML:

- **musi** być deklaratywny,
- **musi** być niezależny od wykorzystywanych protokołów,
- **musi** definiować zestawy standardowych informacji o błędach, które pojawiają się podczas ewaluacji zapytania,
- **nie może**, wykluczać możliwości rozszerzania go o klauzule przeznaczone do definiowania modyfikacji dokumentów XML



Wymagania W3C dotyczące języka zapytań wymagania dotyczące modelu danych

- Wymagania W3C dotyczące modelu danych obejmują następujące kwestie:
 - Oparcie się na informacji XML
 - Typy danych
 - Kolekcje
 - Referencje
 - Dostępność schematu
 - Uwzględnienie przestrzeni nazw

Bazy danych dokumentów XML – wykład 2 – XQuery (3)

Wymagania W3C dotyczące modelu danych XML wykorzystywanego przez zapytania obejmują następujące kwestie: oparcie się na informacji XML, typy danych, kolekcje, referencje, dostępność schematu, uwzględnienie przestrzeni nazw.

Przykładowe ogólne wymagania W3C są następujące:

- Model danych opiera się na informacjach udostępnianych przez procesory XML i procesory schematów i musi gwarantować, że żadna dodatkowa informacja nie będzie wymagana.
- Model danych powinien umożliwiać tworzenie wszystkich elementów informacyjnie istotnych. W przypadku elementów pominiętych powinien udostępniać informacje o powodzie ich pominięcia.
- Model danych musi posiadać reprezentacje zarówno dla ciągów znaków (XML 1.0) jak i prostych i złożonych typów danych zgodnie ze specyfikacją XML Schema.
- Model danych musi posiadać reprezentacje kolekcji dokumentów oraz kolekcji wartości prostych i złożonych.
- Model danych musi wspierać referencje występujące zarówno wewnątrz dokumentu jak i pomiędzy dokumentami.
- Zapytania powinny być możliwe niezależnie od tego czy schematy dla przepytanych dokumentów są dostępne czy też nie.
- Jeżeli schemat jest dostępny, wówczas model danych musi reprezentować każdy z elementów zgodnie z definicją schematu. Dotyczy to dla przykładu domyślnych atrybutów, encji lub typów danych. Elementy te nie są prezentowane jeśli schematy nie istnieją.
- Model danych musi uwzględniać istnienie przestrzeni nazw.



Wymagania W3C dotyczące języka zapytań wymagania dotyczące funkcjonalności języka (1/2)

- Wymagania W3C dotyczące funkcjonalności języka obejmują między innymi następujące zagadnienia:
 - Operacje na typach danych
 - Warunki na elementach tekstowych
 - Hierarchie i sekwencje
 - Kombinacje
 - Agregacje
 - Sortowanie
 - Obsługa wartości pustych

Bazy danych dokumentów XML – wykład 2 – XQuery (4)

Wymagania W3C dotyczące funkcjonalności języka obejmują między innymi następujące zagadnienia: operacje na typach danych, warunki na elementach tekstowych, hierarchie i sekwencje, kombinacje, agregacje, sortowanie, obsługę wartości pustych.

Przykładowe wymagania dotyczące funkcjonalności języka zapytań są następujące:

- Język zapytań musi uwzględniać operacje na wszystkich typach danych dostępnych w ramach modelu danych.
- Zapytania muszą umożliwiać definiowanie prostych warunków dotyczących tekstu i jego zawartości.
- Operacje na kolekcjach muszą wspierać kwantyfikatory ilościowe oraz kwantyfikatory istnienia.
- Zapytania muszą umożliwiać wykonywanie operacji na hierarchiach oraz sekwencjach występujących w strukturze dokumentu.
- Język zapytań przeznaczony do przetwarzania dokumentów XML musi umożliwiać łączenie powiązanej informacji znajdującej się w różnych częściach dokumentu lub wielu różnych dokumentów.
- Musi umożliwiać obliczanie podsumowań w oparciu o grupy elementów.
- Musi umożliwiać porządkowanie generowanych wyników.
- Musi umożliwiać tworzenie wyrażeń, które będą składały się ze złożonych operandów, będących dla przykładu zapytaniami.
- Musi uwzględniać możliwość występowania wartości pustych, dotyczy to również operatorów logicznych.



Wymagania W3C dotyczące języka zapytań wymagania dotyczące funkcjonalności języka (2/2)

- Wymagania W3C dotyczące funkcjonalności języka obejmują między innymi następujące zagadnienia:
 - Zachowywanie struktury
 - Transformacje
 - Referencje
 - Operacje na nazwach elementów
 - Operacje na schematach
 - Rozszerzalność
- Większość wymagań ogólnych, dotyczących modelu danych oraz funkcjonalności języka jest przez XQuery spełniona.

Bazy danych dokumentów XML – wykład 2 – XQuery (5)

Wymagania W3C dotyczące funkcjonalności języka obejmują dodatkowo między innymi następujące zagadnienia: zachowywanie struktury, transformacje, referencje, operacje na nazwach elementów, operacje na schematach, rozszerzalność.

Przykładowe wymagania dotyczące funkcjonalności języka zapytań dotyczące tych zagadnień są następujące:

Zapytania muszą być zdolne do zachowywania (w wyniku) oryginalnej hierarchii lub sekwencji elementów w źródle.

Zapytania muszą umożliwiać transformacje struktur XML, muszą także umożliwiać tworzenie nowych struktur XML.

Zapytania muszą umożliwiać poruszanie się wewnątrz dokumentu, oraz pomiędzy różnymi dokumentami na podstawie referencji.

Zapytania muszą być zdolne do zachowywania tożsamości elementów w ramach modelu danych.

Zapytania muszą umożliwiać wykonywanie prostych operacji na nazwach. Dla przykładu porównywanie nazw, testowanie nazw elementów i atrybutów itp..

Język zapytań powinien umożliwiać wykorzystanie zewnętrznie zdefiniowanych funkcji operujących na dowolnych typach modelu danych. Interfejs takich funkcji powinien być definiowany za pomocą języka zapytań. Powinna być możliwość rozróżnienia funkcji zewnętrznych od funkcji zdefiniowanych w ramach języka zapytań.

Należy w tym momencie zaznaczyć, że znakomita większość wymagań W3C jest przez język XQuery spełniona. Z wymagań funkcjonalnych nie są spełnione wymagania dotyczące operacji na schematach. Wymagania te są następujące:

Zapytania powinny umożliwiać dostęp do schematów XML lub do DTD jeśli tylko one istnieją dla określonego dokumentu.

Jeśli schemat jest zdefiniowany w oparciu o DTD może być wymagane odpowiednie mapowanie na schemat XML.



Cechy języka XQuery

- Język deklaratywny – podobnie jak SQL
- Wykorzystuje wyrażenia ścieżkowe XPath 2.0
- Operuje na pojedynczych dokumentach lub ich kolekcjach
- Konstruuje dokumenty XML lub ich fragmenty
- Oparty na wyrażeniach FLWOR, składających się z pięciu klauzul: **FOR**, **LET**, **WHERE**, **ORDER BY**, **RETURN**

Bazy danych dokumentów XML – wykład 2 – XQuery (6)

Zapytania XQuery pozwalają na definiowanie zapytań w sposób deklaratywny – podobnie jak SQL.

Wykorzystują wyrażenia ścieżkowe zgodne ze standardem XPath 2.0

Operują na pojedynczych dokumentach lub ich kolekcjach. Specjalne funkcje takie jak: input, collection, document lub doc pozwalają na wskazanie danych bazowych, na których zapytanie będzie operowało.

Polecenia XQuery konstruuje dokumenty XML lub fragmenty takich dokumentów – wynik zapytania nie konieczne musi być poprawnym dokumentem XML w sensie well-formed.

Polecenia XQuery oparte są na wyrażeniach nazywanych w skrócie FLWOR. Zbudowane są one z użyciem pięciu klauzul: FOR, LET, WHERE, ORDER BY, RETURN. Na kolejnych slajdach zostaną omówione poszczególne klauzule.



XQuery "w całości" (1/2)

```

for $d in doc("zespoly.xml")//id_zesp
let $e := doc("pracownicy.xml")//pracownik[id_zesp = $d]
where count($e) >= 5
order by avg($e/placa_pod) descending
return
  <duzy-zespol>
  {
    $d,
    <liczba-pracownikow>{count($e)}</liczba-pracownikow>,
    <srednia-placa>{avg($e/placa_pod)}</srednia-placa> }
  </duzy-zespol>

```

Bazy danych dokumentów XML – wykład 2 – XQuery (7)

Przykładowe zapytanie XQuery przedstawione na slajdzie wykorzystuje wszystkie możliwe klauzule. Ich znaczenie w zapytaniu jest następujące

- Klauzula **FOR** wyszukuje w dokumencie zespoly.xml elementy wskazywane przez wyrażenie ścieżkowe //id_zesp. Każdy dopasowany element jest przypisywany zmiennej \$d i powoduje utworzenie tzw. krotki. Klauzula FOR działa podobnie jak klauzule for w innych językach programowania wykorzystuje zmienne do iteracji po określonych wartościach.
- Dla każdego przypisania zmiennej \$d klauzula **LET** przypisuje zmiennej \$e zbiór takich elementów pracownik, które posiadają podelement id_zesp o zawartości identycznej z wartością zmiennej \$d. Elementy pracownik wyszukiwane są z dokumentu o nazwie pracownicy.xml.
- W wyniku działania klauzul FOR i LET otrzymujemy tzw. strumień krotek. Liczba krotek zależy od klauzuli FOR, natomiast liczba zmiennych w krotce zależy od klauzul FOR i LET. W naszym przykładzie liczba krotek jest równa liczbie elementów id_zesp w dokumencie zespoly.xml. Każda krotka zawiera parę zmiennych \$d i \$e posiadających określone wcześniej wartości.
- Zadaniem klauzuli **WHERE** jest selekcja strumienia krotek utworzonych za pomocą klauzul FOR i LET. W naszym przykładzie klauzula WHERE filtruje ten strumień pozostawiając tylko te krotki, w których zbiór elementów pracownik przypisanych do zmiennej \$e składa się, z co najmniej 10 elementów.
- Zadaniem klauzuli **ORDER BY** jest odpowiednie posortowanie wynikowych krotek. W powyższym przykładzie krotki zostaną posortowane wg średniej wartości elementów placa_pod znajdujących się w elementach pracownik przypisanych do zmiennej \$e.
- Na zakończenie klauzula **RETURN** konstruuje wynik zapytania. W naszym przykładzie każda krotka za pomocą klauzuli RETURN stworzy element duzy-zespol, którego zawartością będzie kolejno: element \$d (id_zesp), element liczba-pracownikow którego zawartością będzie liczba będąca ewaluacją wyrażenia count(\$e) oraz element srednia-placa z zawartością wynikającą z wyrażenia avg(\$e/placa_pod)



XQuery "w całości" (2/2)

```
<duzy-zespol>
  <id_zesp>20</id_zesp>
  <liczba-pracownikow>7</liczba-pracownikow>
  <srednia-placa>690,59</srednia-placa>
</duzy-zespol>
```

pracownicy.xml:

```
<?xml version="1.0"?>
<rowset>
  <pracownik>
    <id_prac>130</id_prac>
    <nazwisko>brzezinski</nazwisko>
    <etat>profesor</etat>
    <id_szefa>100</id_szefa>
    <placa_pod>1075.2</placa_pod>
    <placa_dod>611.16</placa_dod>
    <id_zesp>20</id_zesp>
  </pracownik>
  . . .
```

zespoly.xml:

```
<?xml version="1.0"?>
<rowset>
  <zespol>
    <id_zesp>10</id_zesp>
    <nazwa>administracja</nazwa>
  </zespol>
  <zespol>
    <id_zesp>20</id_zesp>
    <nazwa>syst.rozpr.</nazwa>
    <adres>piotrowo 3a</adres>
  </zespol>
  . . .
```

Bazy danych dokumentów XML – wykład 2 – XQuery (8)

Przykładowym wynikiem zapytania z poprzedniego slajdu mógłby być dokument przedstawiony na slajdzie bieżącym w ramce. Dokumenty źródłowe mogłyby mieć natomiast postać przedstawioną poniżej.

Przejdziemy teraz do szczegółowego omówienia poszczególnych klauzul.



Klauzule FOR i LET

- Klauzule FOR i LET wykorzystywane są do tworzenia strumienia krotek. Każda krotka składa się z jednej lub wielu zmiennych.

```
for $s in (<one/>, <two/>, <three/>)
```

```
for $d in document("zespoly.xml")//id_zesp
```

Bazy danych dokumentów XML – wykład 2 – XQuery (9)

Tak jak wspomnieliśmy wcześniej klauzule FOR i LET wykorzystywane są do utworzenia strumienia krotek. Za liczbę krotek odpowiada klauzula FOR. Każdy węzeł dopasowany w wyniku ewaluacji wyrażenia ścieżkowego w klauzuli FOR jest przyporządkowywany odpowiedniej zmiennej użytej w tej klauzuli i stanowi podstawę do utworzenia kolejnej krotki. Zadaniem klauzuli LET jest ewentualne rozszerzenie krotek o zmienne zawierające dodatkowe informacje.

W pierwszym przykładzie ze slajdu klauzula FOR stworzy trzy krotki. W drugim przykładzie stworzy tyle krotek ile elementów `id_zesp` będzie w dokumencie `zespoly.xml`. W obu przypadkach każda krotka będzie składała się tylko z jednej zmiennej.



Klauzula FOR

- Klauzula FOR może zawierać wiele zmiennych
- Klauzula FOR iteruje po każdej wartości uzyskanej w wyniku ewaluacji wyrażenia przypisanego do zmiennej.
- Użycie wielu zmiennych w w klauzuli FOR powoduje, że dla każdej wartości zmiennej nadrzędnej wykonywana jest iteracja za pomocą zmiennej podrzędnej.
- Wynikowa kolejność krotek tworzonych za pomocą klauzuli FOR jest znacząca

```
for $i in (1, 2), $j in (3, 4)
```

```
($i = 1, $j = 3)
($i = 1, $j = 4)
($i = 2, $j = 3)
($i = 2, $j = 4)
```

```
for $z in document("zespoly.xml")//id_zesp,
  $p in document("pracownicy.xml")//pracownicy[id_zesp = $z]
```

Bazy danych dokumentów XML – wykład 2 – XQuery (10)

Klauzula FOR może zawierać wiele zmiennych. Każda zmienna może być związana z odrębnym wyrażeniem.

Klauzula FOR iteruje po każdej wartości uzyskanej na podstawie ewaluacji wyrażenia przypisanego do zmiennej. Oznacza to, że jeżeli wyrażenie przypisane do zmiennej \$x wskazuje na 10 wartości w dokumencie źródłowym, to klauzula FOR utworzy 10 iteracji, w każdej z nich do zmiennej \$x zostanie przypisana kolejna wartość z dokumentu źródłowego.

Użycie wielu zmiennych ma podobny skutek do zagnieżdżenia pętli w językach programowania. Dla każdej wartości zmiennej nadrzędnej wykonywana jest iteracja dla zmiennej podrzędnej.

Dla przykładu w pierwszym zapytaniu na slajdzie mamy dwie zmienne \$i i \$j. Zmienna \$i jest zmienną nadrzędną – wystąpiła wcześniej w klauzuli FOR, natomiast \$j jest zmienną podrzędną. A zatem dla każdego z dopasowań zmiennej \$i zostanie wykonana iteracja przy użyciu zmiennej \$j. Wyrażeniem przypisanym zmiennej \$i jest (1, 2), natomiast wyrażeniem przypisanym zmiennej \$j jest (3, 4). W efekcie wynikiem będzie strumień czterech krotek w postaci przedstawionej po prawej stronie zapytania.

W wyrażeniach przypisanych do zmiennych podrzędnych można używać zmiennych nadrzędnych tak jak ma to miejsce w przypadku drugiego zapytania umieszczonego na slajdzie. Zmiennej \$z przypisane jest wyrażenie doc("zespoly.xml")//id_zesp. A zatem dla każdego elementu id_zesp w dokumencie zespoly.xml zostanie wykonana iteracja przy użyciu zmiennej \$z. Zmiennej \$p natomiast przypisane zostało wyrażenie, które wykorzystuje zmienną \$z. W naszym przypadku dla każdej iteracji z użyciem zmiennej \$z zostanie wykonana iteracja z użyciem zmiennej \$p, której zostaną przypisywane elementy pracownik posiadające podelement id_zesp identyczny z wartością zmiennej \$z.

Wynikowa kolejność krotek tworzonych przez klauzulę FOR jest znacząca, co oznacza, że nie może być zmieniona samoistnie. Do zmiany kolejności krotek służy klauzula ORDER BY.



Klauzula LET

- Klauzula LET może zawierać jedną lub wiele zmiennych.
- Wiązanie zmiennych odbywa się bez iteracji.
- Wartością zmiennej użytej w klauzuli LET są wszystkie węzły (las węzłów) wskazywane przez wyrażenie przypisane do zmiennej
- Zmienne z klauzuli LET są dodawane do krotek wygenerowanych przez klauzulę FOR.
- Jeśli klauzuli FOR nie ma, wówczas LET generuje tylko jedną krotkę

```
let $i := (1, 2), $j := (3, 4)
return <wynik>i={$i}, j={$j} </wynik>
```

```
<wynik>i=1 2, j=3 4</wynik>
```

```
for $e in distinct-values(doc('pracownicy.xml')//etat/text())
let $l := doc('pracownicy.xml')//pracownik[etat = $e]
return <etat nazwa="{ $e }">{count($l)}</etat>
```

```
<etat nazwa="adiunkt">2</etat>
<etat nazwa="asystent">3</etat>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (11)

Klauzula LET także może zawierać jedną lub wiele zmiennych. Różnica w stosunku do FOR polega na tym, że wiązanie zmiennych odbywa się bez iteracji na podstawie wyrażenia przypisanego zmiennej. Innymi słowy wartością zmiennej użytej w klauzuli LET są wszystkie węzły wskazywane przez wyrażenie przypisane do zmiennej.

Przykładowo pierwsze zapytanie na slajdzie wygeneruje wynik przedstawiony po jego prawej stronie. Zwróćmy uwagę na to, że nie doszło do żadnych iteracji. Do zmiennych \$i i \$j zostały przypisane odpowiednio wyrażenia 1,2 i 3,4.

Wynika to z faktu, iż zarówno do zmiennej \$i jak i \$j przypisany został las węzłów określony na podstawie wyrażen przypisanych zmiennym.

Zwróćmy uwagę, że w omawianym zapytaniu nie występuje klauzula FOR. Takie zapytania oczywiście są możliwe i generują zawsze jedną krotkę.

Zmienne z klauzuli LET są dodawane do krotek wygenerowanych przez klauzulę FOR.

Przeanalizujmy, przykładowo drugie zapytanie istniejące na slajdzie.

Klauzula FOR utworzy krotki ze zmiennymi \$e, których wartością będą ciągi znaków umieszczone w elementach etat w dokumencie pracownicy.xml. Krotek będzie tyle ile różnych wartości tekstowych w elementach etat w dokumencie pracownicy.xml. Dla każdej wartości zmiennej \$e, klauzula LET przypisze do zmiennej \$l las elementów pracownik, posiadających w podelemencie etat wartość zgodną z ciągiem znaków przypisanym do zmiennej \$e. W efekcie ewaluacji powyższego zapytania możemy otrzymać wynik w postaci przedstawionej na dole slajdu.



FOR vs LET

- Mimo iż obie klauzule FOR i LET służą do wiązania zmiennych, sposób w jaki to robią jest zupełnie różny.

```
for $s in (<one/>, <two/>, <three/>)  
return <out>{$s}</out>
```

```
let $s := (<one/>, <two/>, <three/>)  
return <out>{$s}</out>
```

```
<out><one/> </out>  
<out><two/> </out>  
<out><three/> </out>
```

```
<out><one/><two/><three/></out>
```

Mimo iż obie klauzule FOR i LET służą do wiązania zmiennych, sposób, w jaki to robią, jest zupełnie różny. Dobrze podsumowują ten fakt zapytania przedstawione na aktualnym slajdzie.

Lewe zapytanie skonstruuje trzy krotki. Do zmiennej \$s w trzech kolejnych iteracjach zostaną przypisane trzy kolejne elementy z wyrażenia przypisanego zmiennej. Pod zapytaniem przedstawiono jego wynik.

Zapytanie przedstawione na slajdzie po prawej stronie skonstruuje tylko jedną krotkę. Wartością zmiennej \$s będzie las elementów pochodzący z wyrażenia przypisanego zmiennej. Poniżej przedstawiono wynik tego zapytania.



Klauzula FOR i zmienne pozycji

- Zmienna pozycji (positional variable) może wystąpić w klauzuli FOR i musi być poprzedzona słowem kluczowym AT.
- Kolejne iteracje przypisują zmiennym pozycji kolejne numery porządkowe.
- Wartości zmiennych pozycji rozpoczynają się od 1, a ich typem jest xs:integer

```
for $scar at $i in ("Ford", "Chevy"),
    $pet at $j in ("Cat", "Dog")
return <wynik>{$scar} at {$i}, {$pet} at {$j} </wynik>
```

```
<wynik>Ford at 1, Cat at 1</wynik>
<wynik>Ford at 1, Dog at 2</wynik>
<wynik>Chevy at 2, Cat at 1</wynik>
<wynik>Chevy at 2, Dog at 2</wynik>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (13)

Zmienna pozycji (ang. *positional variable*) może wystąpić w klauzuli FOR i musi być poprzedzona słowem kluczowym AT.

Kolejne iteracje przypisują zmiennym pozycji kolejne numery porządkowe.

Wartości zmiennych pozycji rozpoczynają się od wartości 1, a ich typem jest xs:integer.

Przykładowo zapytanie, które zostało przedstawione na slajdzie przyporządkuje zmiennym \$i kolejno wartości 1 i 2, analogiczne wartości zostaną przypisane zmiennej \$j. Wynik powyższego zapytania będzie miał postać przedstawioną poniżej zapytania.

Wartości zmiennych pozycji są niezależne od ewentualnego sortowania. I tak, gdyby przykładowo do naszego zapytania dodać klauzulę ORDER BY sortującą ostateczny wynik względem rosnącej wartości zmiennej \$j, a następnie względem wartości zmiennej \$scar, to wygenerowany wynik byłby identyczny z przedstawionym na slajdzie z dokładnością do kolejności krotek.

Oczywiście zmienne pozycji mogą być wykorzystywane tak samo jak każde inne zmienne. Zwiększają one liczbę zmiennych w każdej z krotek generowanych przez klauzule FOR



Klauzula WHERE

- Opcjonalna klauzula WHERE służy do filtrowania krotek
- Wyrażenie zdefiniowane w klauzuli WHERE jest obliczane dla każdej z krotek.
- Jeśli wyrażenie jest prawdziwe, to krotka jest poddawana dalszemu przetwarzaniu, np. w klauzuli RETURN. W przeciwnym wypadku krotka jest odrzucana.

```
avg(for $x at $i in (1,2,3,4,5,6,7,8,9,10)
  where $i mod 2 = 0
  return $x)
```

```
for $e in distinct-values(doc('pracownicy.xml')//etat/text())
let $l := doc('pracownicy.xml')//pracownik[etat = $e]
where count($l) > 2
return <etat nazwa="{ $e }">{count($l)}</etat>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (14)

Podobnie jak klauzule FOR i LET klauzula WHERE jest opcjonalna.

Zadaniem klauzuli WHERE jest selekcja krotek wygenerowanych za pomocą klauzul FOR i LET, czyli zaprzestanie przetwarzania (odrzućenie) krotek, które nie spełniają zadanego warunku.

Wyrażenie zdefiniowane w klauzuli WHERE powinno być wyrażeniem logicznym i jest obliczane dla każdej z krotek. Jeśli wyrażenie jest prawdziwe, to krotka poddawana jest dalszemu przetwarzaniu, np. w ramach klauzuli RETURN lub ORDER BY. W przeciwnym wypadku krotka jest odrzucana.

Przykładowo klauzula WHERE w podzapytaniu umieszczonym we wnętrzu funkcji AVG (zapytanie to zostało przedstawione na slajdzie jako pierwsze) z dziesięciu krotek wygenerowanych przez klauzulę FOR, pozostawi tylko pięć, które dzięki klauzuli RETURN zostaną przekazane do dalszego przetwarzania za pomocą funkcji AVG. W naszym przypadku wartości te będą musiały spełnić warunek podzielności bez reszty przez 2, a zatem będą to wartości 2,4,6,8,10.

Zwróćmy uwagę, że w zapytaniu na poziomie funkcji AVG nie występują żadne z omówionych wcześniej klauzul.

Innym przykładem może być zapytanie przedstawione na slajdzie jako drugie.

W tym drugim przypadku elementy etat zostaną utworzone przez klauzulę RETURN tylko w przypadku, gdy na danym etacie znajduje się więcej niż dwóch pracowników. Odrzucenie pozostałych etatów – krotek jest realizowane za pomocą klauzuli WHERE weryfikującej liczbę elementów przypisanych do zmiennej \$l.



Klauzula RETURN i ORDER BY

- Klauzula RETURN jest ewaluowana raz dla każdej krotki uzyskanej w wyniku działania wcześniejszych klauzul. Wyniki tej ewaluacji są składają się na rezultat wyrażenia FLWOR.
- Jeśli w zapytaniu nie ma klauzuli ORDER BY wówczas kolejność krotek dostarczanych do klauzuli RETURN jest determinowana przez sposób przetwarzania klauzul FOR i LET.
- Jeśli klauzula ORDER BY istnieje wówczas to ona decyduje o kolejności krotek w strumieniu dostarczonym do klauzuli RETURN.

Bazy danych dokumentów XML – wykład 2 – XQuery (15)

Zadaniem klauzuli RETURN jest konstruowanie wyniku zapytania XQuery. Wynik może mieć postać dokumentów XML (poprawnych w sensie well-formed) lub ich fragmentów (nie spełniających reguł poprawności w sensie well-formed) .

Klauzula RETURN jest ewaluowana raz, dla każdej krotki uzyskanej w wyniku działania wcześniejszych klauzul. Wyniki tej ewaluacji składają się na rezultat wyrażenia FLWOR.


Jeśli w zapytaniu nie ma klauzuli ORDER BY wówczas kolejność krotek dostarczanych do klauzuli RETURN jest determinowana przez sposób przetwarzania klauzul FOR i LET.

Jeśli klauzula ORDER BY istnieje wówczas to ona decyduje o kolejności krotek w strumieniu dostarczonym do klauzuli RETURN.



Opcjonalność klauzuli

```
avg(for $x at $i in (1,2,3,4,5,6,7,8,9,10)
     where $i mod 2 = 0
     return $x)
```



```
let $q := avg(for $x at $i in (1,2,3,4,5,6,7,8,9,10)
              where $i mod 2 = 0
              return $x)
return $q
```

Zapytania XQuery mogą nie posiadać żadnej z omawianych klauzul.

Jako przykład przeanalizujemy wcześniejszy przykład z użyciem funkcji AVG. (został on ponownie przedstawiony na bieżącym slajdzie). Zapytanie na poziomie funkcji AVG nie posiada żadnej zmiennej, żadnej klauzuli. W przypadku takich zapytań zakłada się, że wyrażenie XQuery będące treścią zapytania zostaje przypisane za pomocą niejawnej klauzuli LET do niejawnej zmiennej, która następnie jako jedyna użyta jest w niejawnej klauzuli RETURN.

Innymi słowy zapytanie przedstawione w górnej części slajdu jest skrótem zapytania przedstawionego poniżej.



Klauzula ORDER BY

- Zadaniem klauzuli ORDER BY jest uszeregowanie strumienia krotek
- Klauzula ORDER BY definiuje porządek w oparciu o jedno lub wiele wyrażeń.
- Słowa kluczowe występujące w wyrażeniach klauzuli ORDER BY:
 - descending – porządek sortowania malejący
 - empty least – wartości puste traktowane są jako najmniejsze
 - empty greatest – wartości puste traktowane są jako największe

```
for $p in doc("pracownicy.xml")//pracownik
order by number($p/placa_pod) descending
return <w>{$p/nazwisko} {$p/placa_pod}</w>
```

```
for $z in doc("zespolny.xml")//zespol
order by $z/adres empty greatest
return $z
```

Bazy danych dokumentów XML – wykład 2 – XQuery (17)

Zadaniem klauzuli ORDER BY jest uszeregowanie strumienia krotek wg określonego porządku. Porządek ten definiuje się w sposób bardzo podobny jak to jest przypadku klauzuli ORDER BY w języku SQL.

Klauzula ORDER BY może zawierać definicje porządku opartą na jednym lub wielu wyrażeniach.

Porządek wyznaczany jest poprzez sortowanie krotek w oparciu o wyrażenia kolejno od lewej do prawej.

Słowa kluczowe, jakie mogą wystąpić przy wyrażeniach znajdujących się w klauzuli ORDER BY to:

descending – ustala porządek sortowania malejący, domyślnym porządkiem jest porządek rosnący

empty least – powoduje, że wartości puste ustawiane są na końcu

empty greatest – w tym przypadku wartości puste traktowane są jako największe

stable – gwarantuje zachowanie oryginalnego porządku krotek (wynikającego z przetwarzania klauzul FOR i LET) w przypadku takiej samej wartości wyrażenia. W przeciwnym wypadku kolejność może być dowolna np.. zależna od sposobu realizacji operacji sortowania.

Przykładowo zapytanie przedstawione na slajdzie jako pierwsze, uszereguje strumień krotek generowany przez klauzule FOR i LET zgodnie z malejącymi wartościami elementów placa_pod. Natomiast zapytanie przedstawione na slajdzie jako drugie, uszereguje strumień krotek alfabetycznie względem elementu adres, przy czym wartości puste zostaną potraktowane jako największe.



Uporządkowanie wyników

- Wyrażenia XQuery zwracają wynik w ściśle określonym porządku
 - wyznaczonym przez klauzulę ORDER BY
 - wyznaczonym przez kolejność węzłów w dokumencie odczytywanych za pomocą klauzuli FOR
- Deklaracja pozwalająca na zignorowanie porządku poszczególnych elementów ma postać funkcji unordered

```
unordered(for $z in doc("zespoly.xml")//id_zesp,
           $p in doc("pracownicy.xml")//pracownik
           where $p/id_zesp = $z
           return <pracownik>{$p/nazwisko},{ $z/ ../nazwa}</pracownik>
           )
```

Bazy danych dokumentów XML – wykład 2 – XQuery (18)

W ogólnym przypadku wyrażenia XQuery zwracają wynik w ściśle określonym porządku. Wynika on albo z wyrażeń umieszczonych w klauzuli ORDER BY albo z oryginalnego porządku węzłów odczytywanych za pomocą klauzuli FOR.

Mimo iż dokumenty XML stanowią uporządkowane zbiory tekstowe, użytkownika może nie interesować kolejność zwracanych przez zapytanie XQuery elementów. Ze względów efektywnościowych można, zatem pozwolić sobie na pobieranie poszczególnych elementów bez zachowywania ich oryginalnego porządku.

Korzyści z takiego rozwiązania mogą być znaczące dla przykładu w sytuacjach, gdy dokumenty w rzeczywistości składowane są dla przykładu w relacyjnych lub obiektowych bazach danych i odtwarzanie dokumentów XML z uwzględnieniem oryginalnej kolejności elementów może być operacją kosztowną.

Deklaracja pozwalająca na zignorowanie porządku wynikowego ma postać funkcji unordered. Argumentem funkcji jest zapytanie, dla którego wynikowe krotki mogą być uzyskane w dowolnej kolejności.

Przykładowe zapytanie z wykorzystaniem funkcji unordered zostało przedstawione na slajdzie. Gdyby funkcji unordered nie było zbiór wynikowych elementów pracownik powinien się rozpoczynać od pracowników należących do pierwszego zespołu umieszczonego w dokumencie zespoly.xml. Użycie funkcji unordered może dla przykładu spowodować wykonanie operacji połączenia informacji pochodzącej z dokumentu zespoly.xml i pracownicy.xml w dowolny sposób, a następnie wygenerowanie odpowiedzi bez dodatkowego sortowania.



Podzapytania

- W zapytaniach XQuery można zagnieżdżać wyrażenia FLWOR
- Najczęściej zagnieżdżenia występują
 - w klauzuli RETURN
 - jako argumenty funkcji
 - w klauzuli WHERE pod postacią np. wyrażeń ilościowych
- Zagnieżdżenia mogą wykorzystywać wcześniej przypisane zmienne

```
for $z in doc("zespoly.xml")//id_zesp
return <zespół> {$z/../nazwa}
  {for $p in doc("pracownicy.xml")//pracownik
   where $p/id_zesp = $z
   return $p/nazwisko}
</zespół>
```

```
<wynik>
{
  for $z in doc("zespoly.xml")//nazwa
  return $z }
</wynik>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (19)

W zapytaniach XQuery można zagnieżdżać wyrażenia FLWOR.

Najczęściej zagnieżdżenia występują:

- w klauzuli RETURN
- jako argumenty funkcji
- w klauzuli WHERE pod postacią np. wyrażeń ilościowych

Zagnieżdżenia mogą wykorzystywać wcześniej przypisane zmienne. Zagnieżdżenia w klauzuli RETURN, umieszczone pomiędzy ciągami znaków tworzącymi szablon wyniku, ograniczamy nawiasami klamrowymi. W podobny sposób postępujemy z wymagającymi ewaluacji odwołaniami do wartości zmiennych.

Przykładowe zapytania z zagnieżdżonym wyrażeniem FLWOR umieszczono na slajdzie.

W pierwszym przypadku zapytanie zewnętrzne odwołuje się do dokumentu zespoly.xml i wydobywa z niego elementy id_zesp, które przypisuje do zmiennej \$z, każdorazowo tworząc nową krotkę. Dla każdej krotki klauzula RETURN tworzy element zespół, w którego wnętrzu zostanie umieszczony element nazwa (ewaluacja zmiennej wymaga użycia nawiasów klamrowych) oraz wynik zagnieżdżonego wyrażenia FLWOR – las elementów nazwisko. W podzapytaniu wykorzystano zewnętrzną zmienną \$z do ograniczenia krotek (wydobycia pracowników należących do danego zespołu) co oczywiście skutkuje tym że podzapytanie utworzy odpowiednio wyselekcjonowane elementy nazwisko.

Podzapytanie bywa też najczęstszym sposobem na generowanie wyników w postaci dokumentów poprawnych w sensie well-formed. Zapytanie drugie jest tego przykładem.



Wyrażenia warunkowe

- XQuery wspiera wyrażenia warunkowe oparte na konstrukcji IF, THEN, ELSE
- W przypadku, gdy wyrażenie testowane w warunku występującym po IF jest spełnione, wynikiem wyrażenia warunkowego jest wyrażenie występujące po THEN. W przeciwnym przypadku, wynikiem jest wyrażenie występujące po ELSE.

```
for $z in doc("zespol.xml")//id_zesp
return <zespól> {$z/../nazwa}
{ if ($z = 10) then
  for $p at $i in doc("pracownicy.xml")//pracownik
  where $p/id_zesp = $z
  return $p/nazwisko
  else () }
</zespól>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (20)

XQuery wspiera także wyrażenia warunkowe oparte na konstrukcji IF, THEN, ELSE. Wykorzystuje się je najczęściej do warunkowego tworzenia fragmentów wyników.

W przypadku, kiedy wyrażenie testowane w warunku występującym po słowie kluczowym IF jest spełnione, wynikiem wyrażenia warunkowego jest wyrażenie występujące po słowie kluczowym THEN. W przeciwnym przypadku wynikiem wyrażenia warunkowego jest wyrażenie występujące po słowie kluczowym ELSE.

Przykład zapytania z użyciem klauzuli IF THEN ELSE został przedstawiony na slajdzie. W zapytaniu tym nazwiska pracowników są generowane tylko dla zespołu o identyfikatorze 10-tym. Dla pozostałych zespołów element wynikowy zespól będzie składał się tylko z jednego podelementu nazwa.

Zwróćmy uwagę na obowiązkową kompletność wyrażenia IF, THEN, ELSE (konieczność użycia wszystkich słów kluczowych)



Wyrażenia ilościowe (1/2)

- Wyrażenia ilościowe wspierają weryfikację warunków dla zbioru elementów.
- Wyrażenia ilościowe są spełnione gdy warunek testujący jest spełniony dla
 - wszystkich elementów w zbiorze
 - chociaż jednego elementu w zbiorze
- Format wyrażenia ilościowego jest następujący:

```
(some|every) var in expr (, var2 in expr2)* satisfies WarunekTestujacy
```

```
for $z in doc("zespoly.xml")//id_zesp
where some $p in doc("pracownicy.xml")//pracownik
      satisfies $p/id_zesp = $z
return $z/../nazwa
```

Bazy danych dokumentów XML – wykład 2 – XQuery (21)

Wyrażenia ilościowe wspierają weryfikację warunków dla zbioru elementów.

Wyrażenia ilościowe są spełnione, gdy warunek testujący jest spełniony dla:

- wszystkich elementów w zbiorze (every),
- chociaż jednego elementu w zbiorze (some).

Format wyrażenia ilościowego został przedstawiony na slajdzie. Po słowie kluczowym satisfies umieszczamy warunek testujący który w zależności od użytego słowa kluczowego (some lub every) będzie musiał być spełniony dla choćby jednego lub dla każdego elementu w zbiorze tworzonym przez wyrażenie ilościowe. Z reguły warunek testujący wykorzystuje zmienne zastosowane w wyrażeniu ilościowym.

Przykład prostego zapytania wykorzystującego wyrażenie ilościowe został zamieszczony na slajdzie. W zapytaniu tym poszukiwane są zespoły posiadające co najmniej jednego pracownika.



Wyrażenia ilościowe (2/2)

```
for $z in doc("zespoly.xml")//id_zesp
where every $p in doc("pracownicy.xml")//pracownik[id_zesp = $z]
      satisfies $p/placa_pod > 500
return $z/../../nazwa
```

```
<frequent_bidder>
{ for $u in doc("users.xml")//user_tuple
  where
    every $item in doc("items.xml")//item_tuple satisfies
      some $b in doc("bids.xml")//bid_tuple satisfies
        ($item/itemno = $b/itemno and $u/userid = $b/userid)
  return
    $u/name }
</frequent_bidder>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (22)

Aby bliżej zapoznać się z możliwościami wyrażeń ilościowych przeanalizujemy dwa kolejne zapytania.

W pierwszym zapytaniu chcemy poznać nazwę zespołów, w których wszyscy pracownicy zarabiają powyżej 500 zł.

Nieco bardziej złożonym przypadkiem jest drugie zapytanie. Wykorzystuje ono trzy dokumenty:

- users.xml – zawiera on informację o użytkownikach serwisu aukcyjnego
- items.xml – zawiera informacje o przedmiotach wystawionych na aukcjach
- bids.xml – zawiera informacje o poszczególnych ofertach składanych przez użytkowników na rzecz wystawionych przedmiotów.

Przejdźmy teraz do analizy naszego zapytania. Element frequent_bidder został użyty w celu zagwarantowania wyniku poprawnego w sensie well-formed. W jego wnętrzu zapytanie przegląda informacje dotyczące użytkowników z dokumentu users.xml. Klauzula WHERE sprawdza czy dla każdego przedmiotu (o którym informacja istnieje w dokumencie items.xml) użytkownik (dostępny za pomocą zmiennej \$u) złożył chociaż jedną ofertę (informacja o tej ofercie znajduje się w dokumencie bids.xml).

Przykład ten jest bardzo popularny i w znakomity sposób pokazuje możliwości wyrażeń ilościowych.



Funkcje

- Dostępu
- Funkcja błędu
- Funkcje i operatory działające na liczbach
 - Operatory
 - Operatory porównania
 - Funkcje
- Funkcje i operatory działające na ciągach znaków
 - Operatory porównania
 - Funkcje
- Operujące na wartościach logicznych
- Operujące na datach czasie, okresach
 - Operatory porównania
 - Funkcje ekstrakcji
 - Funkcje arytmetyczne na okresach
 - Funkcje stref czasowych
 - Funkcje i operatory na przedziałach czasu
- Operujące na nazwach elementów
- Operujące na węzłach
- Operujące na sekwencjach
- Kontekstu

<http://www.w3.org/TR/xquery-operators/>

Bazy danych dokumentów XML – wykład 2 – XQuery (23)

XQuery jest bardzo bogaty we wszelkiego rodzaju funkcje. Oprócz funkcji predefiniowanych, użytkownik może skorzystać z funkcji definiowanych samodzielnie.

Funkcje predefiniowane można podzielić w następujący sposób:

- Funkcje dostępu
- Funkcja błędu
- Funkcje i operatory działające na liczbach
 - * Operatory
 - * Operatory porównania
 - * Funkcje
- Funkcje i operatory działające na ciągach znaków
 - * Operatory porównania
 - * Funkcje
- Funkcje operujące na wartościach logicznych
- Funkcje operujące na datach czasie, okresach
 - * Operatory porównania
 - * Funkcje ekstrakcji
 - * Funkcje arytmetyczne na okresach
 - * Funkcje stref czasowych
 - * Funkcje i operatory na przedziałach czasu
- Funkcje operujące na nazwach elementów
- Funkcje operujące na węzłach
- Funkcje operujące na sekwencjach
- Funkcje kontekstu



Przykładowe funkcje i operatory działające na ciągach znaków

Funkcja (XPath 1.0)	Wynik	Opis
xf:concat(string? \$op1, string? \$op2, ...)	string	Konkatenacja ciągów znaków
xf:starts-with(string? \$operand1, string? \$operand2, anyURI? \$collationLiteral)	boolean?	Weryfikacja czy operand1 rozpoczyna się od operandu2
xf:ends-with(string? \$operand1, string? \$operand2, anyURI? \$collationLiteral)	boolean?	Weryfikacja czy operand1 kończy się na operandzie2
xf:contains(string? \$operand1, string? \$operand2, anyURI? \$collationLiteral)	boolean?	Weryfikacja czy operand1 zawiera operand2
xf:substring(string? \$sourceString, decimal? \$startingLoc, decimal? \$length)	string?	Z ciągu sourceString wycina od pozycji startingLoc ciąg długości length

Bazy danych dokumentów XML – wykład 2 – XQuery (24)

Na bieżącym slajdzie przedstawiono przykładowe funkcje i operatory działające na ciągach znaków. I tak:

- funkcja concat łączy ciągi znaków przekazane jej jako parametry,
- operator start-with sprawdza czy pierwszy parametr rozpoczyna się od ciągu znaków przekazanego pod postacią drugiego parametru,
- operator ends-with sprawdza czy pierwszy parametr kończy się zestawem znaków przekazanym pod postacią drugiego parametru,
- operator contains sprawdza czy pierwszy parametr zawiera zestaw znaków przekazany jako drugi parametru,
- a funkcja substring pozwala na wycięcie dowolnego fragmentu z podanego ciągu znaków.

Pełen zestaw funkcji wchodzących w zakres standardu XQuery jest przedstawiony na stronie organizacji W3C (<http://www.w3.org/TR/xquery-operators/>)

Na następnym slajdzie zostanie przedstawionych kilka zapytań wykorzystujących funkcje i operatory wchodzące w zakres standardu XQuery.



Przykłady zastosowań funkcji i operatorów

```

for $z in doc("zespoly.xml")//id_zesp
let $p := doc("pracownicy.xml")//pracownik[id_zesp = $z]
where ends-with($z/../nazwa,'e')
return <wynik>
  { if (count($p)>0) then
    $z/../nazwa else
    ( error(<a>Zespól bez pracownikow</a> ) )
  }
</wynik>

```

```

for $w in tokenize("bazy, danych, dokumentów, XML", ",\s*")
where matches($w, '^d.*w$')
return <wynik>{$w}</wynik>

```

Bazy danych dokumentów XML – wykład 2 – XQuery (25)

Zapytania przedstawione na bieżącym slajdzie zawierają przykłady zastosowań funkcji i operatorów.

Zapytanie pierwsze przegląda elementy `id_zesp` znajdujące się w dokumencie `zespoly.xml` i przypisuje je do zmiennej `$z`. Dla każdego znalezionej elementu poszukiwane są elementy `pracownik` posiadające element `id_zesp` o wartości identycznej z wartością zmiennej `$z`. Wszystkie znalezione elementy `pracownik` są przypisywane zmiennej `$p`. Warunek umieszczony w klauzuli `WHERE` został zbudowany w oparciu o operator `ends-with`. Warunek nakłada filtr na generowany strumień krotek – do dalszego przetwarzania zostaną skierowane te zespoły, których nazwa kończy się na literę `e`.

Klauzula `RETURN` w pierwszym zapytaniu wykorzystuje przetwarzanie warunkowe. Jeżeli liczba pracowników w danym zespole jest mniejsza lub równa 0 wówczas wykonywanie zapytania zostanie przerwane i zgłoszony zostanie błąd o treści: `Zespól bez pracownikow`. Zgłoszenie błędu będzie możliwe dzięki zastosowaniu funkcji błędu – `error`.

Drugie zapytanie za pomocą funkcji `tokenize` dzieli ciąg znaków `"bazy, danych, dokumentów, XML"` na fragmenty przy założeniu, że separatorem wyznaczającym poszczególne składowe jest znak `,`. Kolejne fragmenty przypisywane są w kolejnych krotkach do zmiennej `$w`. Wartość zmiennej `$w` w klauzuli `WHERE` jest analizowana za pomocą funkcji `matches`. Funkcja `matches` porównuje zgodność ciągu znaków ze wzorcem. W naszym przypadku sprawdza, czy zawartość zmiennej `$w` rozpoczyna się od litery `d` i kończy na literze `w`, tylko takie fragmenty będą podlegały dalszemu przetwarzaniu za pomocą klauzuli `RETURN`.



Funkcje użytkownika

- Oprócz funkcji wbudowanych użytkownik może definiować i korzystać z funkcji własnych
- Funkcje własne muszą być definiowane i używane w predefiniowanej przestrzeni nazw local
- Funkcje mogą mieć parametry, które mogą być typowane

```

declare function
  local:liczba_pracownikow($param)
  { count(doc("pracownicy.xml")//pracownik[id_zesp=$param]) };
for $z in doc("zespoly.xml")//id_zesp
let $l := local:liczba_pracownikow($z)
return
<zespole>
  {$z/./nazwa}
  <liczba_prac>{$l}</liczba_prac>
</zespole>

```

Bazy danych dokumentów XML – wykład 2 – XQuery (26)

Oprócz funkcji wbudowanych użytkownik może definiować i korzystać z funkcji własnych – lokalnych.

Funkcje lokalne muszą być definiowane i używane w predefiniowanej przestrzeni nazw local.

Funkcje mogą mieć parametry, które mogą być typowane. Nie można przeciążać nazw funkcji użytkownika ani definiować funkcji o zmiennej liczbie parametrów.

Przykładowa definicja funkcji i jej użycie zostało przedstawione na slajdzie.

Funkcja o nazwie liczba_pracownikow ma jeden parametr \$param. Zgodnie z regułą dotyczącą przestrzeni nazw, funkcja ta została zdefiniowana przy użyciu przestrzeni nazw do której odwołaliśmy się za pomocą prefiksu local. Wynikiem funkcji jest wyrażenie, które obliczy liczbę elementów pracownik w dokumencie pracownicy.xml posiadających podelement id_zesp o wartości równej wartości parametru \$param.

Funkcja liczba_pracownikow została użyta w zapytaniu do wygenerowania zawartości elementu liczba_prac.



Predefiniowane przestrzenie nazw

- <http://www.w3.org/2001/XMLSchema> – wykorzystywana przez konstruktory obiektów, przypisana do prefiksu **xs**
- <http://www.w3.org/2006/xpath-functions> – wykorzystywana przez funkcje, przypisana do prefiksu **fn**
- <http://www.w3.org/2005/xqt-errors> – wykorzystywana przez funkcje błędów przypisana do prefiksu **err**.
- Ponadto spotyka się prefiks **op**, który jest wykorzystywany do definiowania operatorów (nie dostępnych bezpośrednio dla użytkownika), a także inne będące lokalnymi rozszerzeniami

Bazy danych dokumentów XML – wykład 2 – XQuery (27)

Oprócz przestrzeni nazw o prefiksie lokal w XQuery istnieje jeszcze kilka predefiniowanych przestrzeni nazw.

Przeźren nazw <http://www.w3.org/2001/XMLSchema> wykorzystywana jest przez konstruktory obiektów i przypisana została do prefiksu **xs**

Przeźren nazw <http://www.w3.org/2006/xpath-functions> wykorzystywana jest przez funkcje wbudowane i przypisana została do prefiksu **fn**

Przeźren nazw <http://www.w3.org/2005/xqt-errors> wykorzystywana jest przez funkcje błędów i przypisana została do prefiksu **err**

Ponadto spotyka się prefiks **op**, który jest wykorzystywany do definiowania operatorów (nie dostępnych bezpośrednio dla użytkownika), a także inne przestrzenie nazw będące lokalnymi rozszerzeniami dodawanymi przez twórców aplikacji w celu rozszerzenia wachlarza dostępnych funkcji. Przykładem może być prefiks **ora** możliwy do wykorzystania podczas wykonywania zapytań XQuery w bazie danych Oracle.



Inne języki zapytań: XML-QL

- Język deklaratywny
- Oparty jest na dwóch składowych: wzorcu i szablonie
 - wzorzec – pełni rolę analogiczną do XPath w języku XQuery oraz klauzul FOR, LET i WHERE, za pomocą wzorca do zmiennych przypisywane są wartości pochodzące z dokumentu źródłowego
 - szablon – pełni rolę analogiczną do klauzuli RETURN w języku XQuery, za pomocą szablonu tworzony był odpowiednio skonstruowany wynik
- Składnia:

<code>WHERE wzorzec</code>
<code>CONSTRUCT szablon</code>

Bazy danych dokumentów XML – wykład 2 – XQuery (28)

Przed językiem zapytań XQuery powstało bardzo wiele propozycji i rozwiązań języków zapytań przeznaczonych do przetwarzania baz danych dokumentów XML. Najbardziej znane z nich to XML-QL i Quilt.

XML-QL jest językiem deklaratywnym, który umożliwia wykonywanie zapytań, konstruowanie, transformację oraz integrację danych XML. XML-QL wspiera zarówno uporządkowany jak i nieuporządkowany widok dokumentu XML. Dodatkowo, XML-QL spełnia większość wymagań nakładanych na języki zapytań dla XML przez konsorcjum W3C (<http://www.w3.org/TR/xquery-requirements>).

XML-QL oparty jest na dwóch składowych: wzorcu i szablonie.

Wzorzec pełni rolę analogiczną do XPath w języku XQuery oraz klauzul FOR, LET i WHERE, za jego pomocą do zmiennych przypisywane są wartości pochodzące z dokumentu źródłowego.

Szablon pełni rolę analogiczną do klauzuli RETURN w języku XQuery. Za pomocą szablonu XML-QL tworzony jest odpowiednio skonstruowany wynik w postaci dokumentów XML lub ich fragmentów.

Polecenia XML-QL składają się z dwóch klauzul: WHERE, po którym następuje wzorzec dopasowywany do dokumentu źródłowego i CONSTRUCT, po którym następuje szablon tworzący obiekt wynikowy. Na następnym slajdzie zostanie przedstawione przykładowe zapytanie w języku XML-QL.



XML-QL – przykład

```
zespoly.xml:
<?xml version="1.0"?>
<zespoly>
  <zespol>
    <id_zesp>10</id_zesp>
    <nazwa>administracja</nazwa>
    <adres>piotrowo 3a</adres>
  </zespol>
  <zespol>
    <id_zesp>20</id_zesp>
    <nazwa>badania op.</nazwa>
    <adres>piotrowo 3a</adres>
  </zespol>
</zespoly>
```

```
<male_zespoly> {
  WHERE <zespoly> <$z>
        <nazwa> $n </>
        <adres> $a </>
        </> </> in "zespoly.xml",
        $z = 'zespol'
  CONSTRUCT
    <NiA>$n $a</> }
</>
```

\$z	\$n	\$a
-----	-----	-----
zespol	administracja	piotrowo 3a
zespol	badania op.	piotrowo 3a

```
<male_zespoly>
  <NiA>administracja piotrowo 3a</>
  <NiA>badania op. piotrowo 3a</>
</>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (29)

W przykładzie zapytania XML-QL zastosowano podzapytanie. Zostało ono ograniczone nawiasami klamrowymi. Jego zadaniem jest skonstruowanie elementu zewnętrznego `male_zespoly` (jest tu użyta niejawna klauzula `CONSTRUCT`). Podzapytanie składa się z wzorca rozpoczynającego się od elementu `<zespoly>`. Wzorec składa się z dwóch części rozdzielonych przecinkiem. Zadaniem pierwszej części wzorca jest wielokrotne dopasowanie się do zawartości dokumentu `zespoly.xml`. W wyniku każdorazowego dopasowania zmienne `$z`, `$n` i `$a` uzyskują odpowiednie wartości. Jednocześnie, w wyniku każdego dopasowania, tworzona jest kolejna krotka podlegająca przetwarzaniu. Przy założeniu, że dokument `zespoly.xml` został przedstawiony na slajdzie, wzorec użyty w zapytaniu utworzy 2 krotki zawierające każdorazowo trzy zmienne. Wartości zmiennych w kolejnych krotkach zostały przedstawione na slajdzie. Druga część wzorca sprawdza czy zawartość zmiennej `$z` to "zespol". Ten fragment wzorca funkcjonuje analogicznie do klauzuli `WHERE` w XQuery. Krotki wygenerowane za pomocą klauzuli `WHERE` zostają następnie przetwarzane za pomocą klauzuli `CONSTRUCT`. W naszym przykładzie tworzy ona elementy `<NiA>`, których zawartość zostanie zbudowana w oparciu o wartości zmiennych `$n` i `$a`.

Zwróćmy uwagę, że wzorec wykorzystywany przez język XML-QL odpowiada klauzulom `FOR` i `LET` oraz wyrażeniom XPath używanym w języku XQuery. Sposób definiowania wzorca powoduje, że zwykle przyjmuje on duże rozmiary, ponadto nie jest on tak funkcjonalny jak wyrażenia XPath.

Podstawowymi wadami XML-QL są:

- brak możliwości korzystania z wyrażeń XPath i funkcji XPath
- brak obsługi przestrzeni nazw
- brak funkcji grupowych



Inne języki zapytań: Quilt

- Język Quilt jest bardzo podobny do języka XQuery.
- Zapytania Quilt zbudowane są również w oparciu o klauzule FOR, LET, WHERE i RETURN:
 - FOR – generuje krotki składające się ze zbioru zmiennych
 - LET – pozwala na uzupełnianie krotek o dodatkowe zmienne
 - WHERE – filtruje krotki generowane za pomocą klauzuli FOR i LET
 - RETURN – tworzy wynik zapytania

Bazy danych dokumentów XML – wykład 2 – XQuery (30)

Język Quilt jest bardzo podobny do języka XQuery. Wynika to z faktu iż XQuery jest następcą języka Quilt, a przy jego tworzeniu brały udział między innymi te same osoby.

Zapytania języka Quilt składają się z analogicznych klauzul.

FOR – generuje krotki składające się ze zbioru zmiennych.

LET – pozwala na uzupełnianie krotek o dodatkowe zmienne.

WHERE – filtruje krotki generowane za pomocą klauzuli FOR i LET.

RETURN – tworzy wynik zapytania.

Twórcy języka XQuery w dokumencie dotyczącym języka XQuery napisali:

XQuery wywodzi się z języka zapytań przeznaczonego do przetwarzania dokumentów XML o nazwie Quilt. Z kolei język Quilt zapożyczył swoje własności z kilku innych języków zapytań.

Ze standardu XPath oraz języka XQL zapożyczone zostały wyrażenia ścieżkowe właściwe dla dokumentów hierarchicznych.

Z języka XML-QL zapożyczone zostało pojęcie zmiennych wiązanych oraz wykorzystanie zmiennych do tworzenia nowych struktur.

Z języka SQL zapożyczona została idea zestawu klauzul opartych na słowach kluczowych, które udostępniają mechanizmy do transformacji danych.

Z języka OQL zapożyczona została koncepcja języka funkcjonalnego złożonego z wielu różnych rodzajów wyrażeń, które mogą być wielokrotnie zagnieżdżane z zachowaniem ogólności.

Ponadto na język Quilt miały wpływ także inne języki do przetwarzania dokumentów XML jak Lorel i YATL.



Quilt – przykład

```
<wynik>
(
  FOR $z IN DISTINCT document("zespoly.xml")//zespol/id_zesp
  RETURN
    <zespol>
      <nazwa_z>
        $z/../nazwa/text()
      </nazwa_z> ,
      <profesorowie>
        (
          FOR $p IN document("p.xml")//pracownik[id_zesp = $z]/nazwisko
          WHERE $p/../etat = 'PROFESOR'
          RETURN $p SORTBY(.)
        )
      </profesorowie> ,
    </zespol> SORTBY(..nazwa)
)
</wynik>
```

Bazy danych dokumentów XML – wykład 2 – XQuery (31)

Już na pierwszy rzut oka widać, że Quilt jest bardzo podobny do języka XQuery.

Powyższe zapytanie we wnętrzu elementu <wynik> generuje elementy <zespol> powstałe w oparciu o zawartość dokumentu "zespoly.xml".

Każdy element <zespol> zawiera element <nazwa_z> oraz element <profesorowie>, którego zawartość wynika z zawartości dokumentu pracownicy.xml.

Klauzule FOR, LET, WHERE i RETURN pełnią w powyższym zapytaniu role analogiczne jak w poleceniach XQuery.



Bibliografia

- Pełna specyfikacja XQuery <http://www.w3.org/TR/xquery/>
- Definicje funkcji i operatorów <http://www.w3.org/TR/xquery-operators/>
- Przykładowe edytory otwarte na XQuery
 - Oracle JDeveloper www.oracle.com/technology/products/jdev/
 - Altova XMLSpy www.altova.com